
8 Rough Footing on a Mohr-Coulomb Material

8.1 Problem Statement

The prediction of collapse loads under steady plastic flow conditions is one that can be difficult for a
numerical model to simulate accurately (Sloan and Randolph 1982). A simple example of a problem
involving steady flow is the determination of the bearing capacity of a footing on an elastic-plastic
soil. The bearing capacity is dependent on the steady plastic flow beneath the footing, thereby
providing a measure of the ability of UDEC to model this condition.

A strip footing is evaluated to demonstrate the capability of UDEC to predict collapse loads and
model plastic flow of intact material. The strip footing has a rough base with a width of 6.0 m, and
is located on a frictionless, cohesive soil that has the following properties:

density (ρ) 1000 kg/m3

shear modulus (G) 100 MPa
bulk modulus (K) 200 MPa
cohesion (c) 10 kPa
friction angle (φ) 0
dilation angle (ψ) 0

8.2 Analytical Solution

The bearing capacity for a strip footing is from the solution to “Prandtl’s wedge,” as given by
Terzaghi and Peck (1967):

q = (2 + π)c

or

q = 5.14c (8.1)

where c is the cohesion of the material, and q is the bearing capacity stress at failure. The solution
is based on the mode of failure, as shown in Figure 8.1.

Collapse Load

q = (2 + )cπ 

Figure 8.1 Prandtl’s wedge problem of a strip footing on a frictionless soil

8.3 UDEC Model

A plane-strain analysis is performed for the strip-footing problem. Half-symmetry is used, and
boundary conditions are applied, as shown in Figure 8.2.

Applied
Velocity
Footing

Figure 8.2 UDEC model boundary conditions

Two model grids are created for this problem. The first is a single-block model, and the second is
a two-block model created with a diagonal construction joint.

The first model grid, shown in Figure 8.3, is composed of 2048 triangular zones in a diametrically
opposed triangular pattern. As discussed in Section 1.2.5 in Theory and Background, this zone
pattern has been demonstrated to provide reasonable accuracy for calculations involving plastic
collapse. This pattern is created with either the command

block zone gen edge 0.625

or the command

block zone gen quad 0.64

The second model grid, shown in Figure 8.4, is created by first dividing the model region into two
blocks with the command

block cut crack (0,0) (20,10) join

In this case, only the block zone generate edge command can be applied because each block only
contains three corners. The command

block zone gen edge 0.625

creates zoning with boundary gridpoints at the same locations as those in the single-block model.
The two-block model contains 3104 zones.

The test is velocity-controlled with a downward velocity of 1.0 × 10−3 m/sec applied to the grid-
points located along the boundary corresponding to the area representing the footing. A zero velocity
is applied in the x-direction to represent the rough footing condition. The gridpoint locations of
the fixed-velocity boundary condition are indicated on the plots in Figures 8.3 and 8.4.

The footing load is calculated in FISH function stripload by summing the y-direction forces at
the footing gridpoints and dividing by the representative footing area. The footing load is monitored
as a history for comparison with the bearing capacity calculated from Eq. (8.1).

Figure 8.3 UDEC zone geometry for strip footing – single-block model

Figure 8.4 UDEC zone geometry for strip footing – two-block model with
diagonal construction joint

8.4 Results and Discussion

Figure 8.5 shows the model conditions at the end of the analysis for the single-block model. The
behavior shown is very close to that expected from Figure 8.1. Figure 8.6 shows a history of the
bearing capacity versus vertical displacement of the footing for the model using block zone generate
quad zoning. The final value of the bearing capacity for the strip footing is 50.6 kPa, giving an
error of 1.66% when compared to the expected value of 51.4 kPa. The results using block zone
generate edge zoning are essentially identical to those using block zone generate quad zoning for
the single-block model.

In the two-block model, the triangular zoning has an irregular pattern (as shown in Figure 8.4).
This introduces kinematic constraints in the plastic flow calculation and results in an excessively
stiff response, as indicated by the bearing capacity history plot in Figure 8.7. The error after 2 cm
of vertical settlement of the footing is over 10%, and is increasing.

This problem is discussed in Section 1.2.5 in Theory and Background, and can be corrected by
applying “nodal mixed discretization” (also described in this section). By adding the command

block zone nodal-mixed-discretization on

after the block zone enerate edge command, nodal mixed discretization is applied to the triangular
zoning. The improved result is shown in Figure 8.8. Now the error in the calculated bearing capacity
is reduced to 0.8%.

This exercise illustrates that whenever plastic failure and collapse of deformable blocks is to be
simulated, the block zone generate quad command (or the block zone generate edge command with
the block zone nodal-discretization on command) should be applied in order to obtain an accurate
solution.

Figure 8.5 Steady state x-velocity contours and velocity vectors at collapse
load for strip footing

Figure 8.6 History of strip footing load; exact solution also shown
– single-block model

Figure 8.7 History of strip footing load; exact solution also shown
– two block model (zoning by GENERATE edge)

Figure 8.8 History of strip footing load; exact solution also shown
– two-block model (zoning by GENERATE edge with SET nodal on)

8.6 Listing of Data File

Example 8.1 PRAN.DAT

model new
;File:pran.dat
;Title:Prandtl’s Wedge Test
; rough footing on cohesive material
block tolerance corner-round-length 0.01
block tolerance minimum-edge-length 0.02
block create polygon 0 0 0 10 20 10 20 0
;
; block zone gen quad zoning for single-block model
block zone gen quad 0.64
;
; block zone gen edge zoning for single-block model and for two-block model
; block zone gen edge 0.625
;
; nodal mixed discretization for two-block model
; set nodat on
;
; material properties
block zone group ’clay’
bl zone cmodel assign mohr-c dens 1E3 bulk 2E8 shear 1E8 coh 1E4 ...

tens 1E10 range group ’clay’
;
; boundary conditinos
bl grid apply velocity-x 0 range position-x -0.1 0.1 position-y -0.1 10.1
bl grid apply velocity-x 0 range position-x 19.9 20.1 position-y -0.1 10.1
bl grid apply velocity-x 0 range position-x -0.1 20.1 position-y -0.1 0.1
bl grid apply velocity-y 0 range position-x -0.1 20.1 position-y -0.1 0.1
bl grid apply velocity-y -0.001 range position-x -0.1 3 position-y 9.9 10.1
bl grid apply velocity-x 0 range position-x -0.1 3 position-y 9.9 10.1
;
; comparison to analytical solution
fish define p_cons

p_xp = block.gp.near(3.12,10.0)
p_xm = block.gp.near(2.50,10.0)
p_y0 = block.gp.near(0.0,10.0)
solution=(2.0 + math.pi)*1e4

end
@p_cons
;
fish define stripload

sum =0.0
ib = block.head

loop while ib # 0
ig = block.gp(ib)
loop while ig # 0

if block.gp.pos.y(ig) > 9.8 then
if block.gp.pos.x(ig) < 3.0 then

ibou=block.gp.boundary.corner(ig) ; index of boundary corner
if(ibou) > 0 then ; exterior boundary

forcey = block.boundary.force.y(ibou) ; total y-force
sum = sum - forcey

endif
endif

endif
ig = block.gp.next(ig)

endloop
ib = block.next(ib)

endloop
x_p = block.gp.pos.x(p_xp)
x_m = block.gp.pos.x(p_xm)
p_load = 2.0 * sum / (x_p + x_m)
y_disp = -block.gp.disp.y(p_y0)
stripload = p_load
err = (p_load-solution)/solution

end
@stripload
fish history @err
fish history @stripload
fish history @solution
fish history @y_disp
block smallstrain
model save ’strip1.sav’
;
block cycle 35000
model save ’strip2.sav’

