


## 2 Open Stoping Using Vertical Retreat

### 2.1 Problem Statement

A distinct element simulation of a large blasthole, open stoping operation is shown to demonstrate the ability of *UDEC* to model sequential mining steps. The model is for a quartzite orebody for which the potential instability in the stope back is to be evaluated. Of particular concern is the stress concentration in the crown pillar after mining of the stope is completed.

The geometry for this example is illustrated in [Figure 2.1](#). A steeply dipping orebody (average dip of  $80^\circ$ ) is analyzed between the 990 m level and the 1190 m level of the mine. A low-angle discontinuous joint set is also oriented at  $10^\circ$  dip, with average spacing of 30 m. The average thickness of the orebody is 14 m. The upper stope, above the 1090 m level, is mined first; then, mining of the lower level is completed, leaving a 10 m crown pillar.



**Figure 2.1** Initial geometry for blasthole open stoping operation

Four rock types are defined for the analysis: hanging wall quartzite, footwall quartzite, banded ore and a weaker schistose ore. Based on the average laboratory test values, the following properties for these rock types were assumed.

**Table 2.1** *Properties of rock types*

	Hanging Wall	Banded Ore	Schistose Ore	Footwall
Young's modulus (GPa)	62	56	40	67
Poisson's ratio	0.29	0.28	0.33	0.28
Unconfined compressive strength (MPa)	186	168	96	198


Several joint properties were estimated:

joint normal stiffness	5 GPa/m
joint shear stiffness	5 GPa/m
joint friction angle	27°
joint cohesion	0

The pre-mining state of stress was estimated to be hydrostatic. The stress is 33 MPa at the 1190 m level.

## 2.2 UDEC Analysis

The *UDEC* model was created with constitutive models and properties assigned to the blocks using the **block zone cmodel assign** command. This facilitates the assignment of the different rock types to the model. Figure 2.2 shows the different cohesion properties that were assigned to the hanging wall, footwall, banded ore and schistose ore.




**Figure 2.2** Cohesion properties assigned to the UDEC model


The model was first consolidated at the initial stress state by applying a stress boundary condition. This boundary condition was then replaced with a boundary-element boundary to represent an infinite elastic medium in the far field.

After model consolidation, the mining progressed in five stages. First, the upper level blocks were removed for a stope height of 45 m. The lower stope was then mined in four stages of 17 m, 15 m, 15 m and 18 m, leaving the 10 m crown pillar. The final stress concentration is depicted in Figure 2.3. At this stage, most of the stress is transferred to the abutments. The stress build-up in the crown pillar is shown in the stress history plot in Figure 2.4.

Although backfilling was not simulated in this example, the model can simulate backfill emplacement after excavation, and stress histories in the backfill can be monitored like they were for the crown pillar (see Figure 2.4).



**Figure 2.3** *Principal stresses at end of mining sequence*



**Figure 2.4** *Changes in xx- and yy-stresses in crown pillar with mining stages  
History 2: xx-stress, History 3: yy-stress (note: compressive stresses are negative)*

## 2.3 Listing of Data File

### *Example 2.1 STOPE.DAT*

---

```

;File:STOPE.dat
model title "Open Stope Using Vertical Retreat"
model new
;
;establish initial geometry
block tolerance corner-round-length 0.2
block create polygon 0 0 0 200 150 200 150 0
;
block cut joint-set angle 80 trace 300 spacing 14 origin 0 0
block delete range area 0 1210.0
block cut crack 67.864 0 104.129 200
;
;put in discontinuous cross joints
block cut joint-set angle 350 trace 14 gap 14 spacing 30 origin 0 0
block cut joint-set angle 350 trace 14 gap 14 spacing 30 origin 49.48 200
;
;put in joints needed for later excavations
block cut crack 74 100 89 100
block cut crack 84 155 99 155
block cut crack 63 35 78 35
block cut crack 65 52 81 52
block cut crack 68 67 83 67
block cut crack 71 82 86 82
;
; create finite difference triangles in all blocks
;block zone gen edge 100
block zone gen quad 100.0
block zone gen edge 100.0
;
;assign material properties
;hanging wall rock
block zone group 'hangwall'
;banded ore
bl zone group 'bandedore' range region 56.86 0 92.13 200 104.13 200 67.86 0
;schistose ore
blo zone group 'schisore' range region 67.86 0 104.13 200 106.3 200 71.05 0
;foot wall
block zone group 'footwall' range region 71.05 0 106.3 200 160 200 160 0
;
bl zone cmodel assign mohr-c dens 0.002 bulk 4.8E4 shear 2.4E4 coh 93 ...
    range group 'hangwall'
bl zone cmodel assign mohr-c dens 0.002 bulk 4.2E4 shear 2.2E4 coh 84 ...

```

---

```

    range group 'bandedore'
bl zone cmodel assign mohr-c dens 0.002 bulk 3.8E4 shear 1.5E4 coh 48 ...
    range group 'schisore'
bl zone cmodel assign mohr-c dens 0.002 bulk 5.1E4 shear 2.6E4 coh 99 ...
    range group 'footwall'
;
;joint
block contact group 'jmat1'
bl contact cmodel assign area st-s 4E3 st-n 5E3 fric 27 ...
    range group 'jmat1'
; new contact default
block contact cmodel default area st-s 4000 st-n 5000 friction 27
;
;specify stress field
block edge apply stress -33.0 0.0 -33.0 ...
    gradient-x 0.0 0.0 0.0 gradient-y 0.02 0.0 0.02
block insitu stress -33.0 0.0 -33.0 ...
    gradient-x 0.0 0.0 0.0 gradient-y 0.02 0.0 0.02 ...
    stress-ZZ -33.0 gradient-z 0.0 0.02
model gravity 0 -10
block mechanical history solve-ratio
block zone history stress-xx 80.0 105.0
block zone history stress-yy 80.0 105.0
block solve ratio 1.0E-5
model save 'st1.sav'
;
block boundary-element gen range position-x -1 160 position-y -1 201
block boundary-element fix -100 -100 260 300
block boundary-element property density 0.002 bulk 44000 shear 26000
block boundary-element stiff
;
;begin excavation sequence with upper stope
block delete range pos-x 78 98 pos-y 110 155
;
block solve ratio 1.0E-5
model save 'st2.sav'
;
;excavate lowest part of lower stope
block delete range pos-x 63 81 pos-y 35 52
;
block solve ratio 1.0E-5
model save 'st3.sav'
;
;excavate next part of lower stope
block delete range pos-x 65 83 pos-y 52 67
block solve ratio 1.0E-5

```

```
model save 'st4.sav'  
;  
;excavate next part of lower stope  
block delete range pos-x 67 86 pos-y 67 82  
block solve ratio 1.0E-5  
model save 'st5.sav'  
;  
;excavate last part of lower stope  
block delete range pos-x 71 89 pos-y 82 100  
block solve ratio 1.0E-5  
model save 'st6.sav'
```

---

