


# 1 Seismic-Induced Groundfall

## 1.1 Problem Statement

A demonstration simulation of a seismic-induced groundfall is presented to illustrate the use of *UDEC* for analyzing this type of problem. The model shown in Figure 1.1 was developed based on the configuration and dimensions of the 34-1-554 over-cut shown on a section drawing for Fraser Mine, Falconbridge Limited, Sudbury, Ontario. A two-dimensional, plane-strain representation was chosen normal to the axis of the over-cut. The over-cut was modeled as being 5 m high and 10 m wide.



**Figure 1.1** *UDEC model for seismic-induced groundfall*

It was assumed that two continuous joint sets intersect the plane of analysis: one with an orientation of  $45^\circ$ , and the other with an orientation of  $-9^\circ$ . Both sets have a joint spacing of 5 m. For demonstration purposes, a near vertical “artificial” joint was also added to the block in the roof of the excavation to enhance the instability.

From the average laboratory test values provided for the intact rock, several material properties were assumed for the rock blocks:

density	3000 kg/m <sup>3</sup>
Young's modulus	75,000 MPa
Poisson's ratio	0.18

The blocks were assumed to behave elastically only. Coulomb slip behavior was assumed for the joints, and typical textbook values were chosen for joint properties:

joint normal stiffness	20,000 MPa/m
joint shear stiffness	20,000 MPa/m
friction angle	30°
cohesion	0


The in-situ stress state was estimated to be isotropic at 24 MPa (assuming vertical loading due to overlying rock at a depth of approximately 800 m).

## 1.2 UDEC Analysis


The *UDEC* modeling sequence was performed in three stages. First, the model without the over-cut excavation was consolidated under the in-situ stresses. Next, the excavation was introduced and the model cycled to an equilibrium state. The stress distribution around the over-cut at this stage is illustrated in [Figure 1.2](#). The blocks immediately above and below the over-cut have slipped and then stabilized.

In the third stage, two different seismic events with different peak velocities were evaluated. For all seismic simulations, viscous boundaries were introduced around the outer perimeter of the problem domain to eliminate wave reflections, thereby simulating an infinite rock mass. Seismic events were represented by a sinusoidal y-directed stress wave applied at the top of the model. The applied stress wave was superimposed on the existing in-situ stresses.


In the first simulation, a peak stress of 1.25 MPa was applied. Note that, due to the viscous boundary conditions in effect at the top of the model, the “effective” applied stress is 1.25 MPa/2, or 0.625 MPa. The stress distribution in the roof of the excavation after 0.02 second is shown in [Figure 1.3](#). Displacements were monitored at two points. Point 1 is located in the left corner of the excavation; Point 2 is located at the right corner of the roof block. Displacement versus time plots ([Figure 1.4](#)) for these points essentially show an elastic response.



**Figure 1.2** Stress distribution around excavation at end of excavation stage



**Figure 1.3** Stress distribution in roof of excavation after 0.02 sec. (applied stress =  $1.25 \times \cos(2\pi 100t)$ )




**Figure 1.4** *y-displacement histories for two points on excavation boundary (applied stress =  $1.25 \times \cos(2\pi 100t)$ )*

It is interesting to compare estimated applied velocities with calculated velocities at the top of the model. The following equation can be used to estimate the applied velocity.

$$V = \frac{\sigma}{2 (\rho C_p)} \quad (1.1)$$

where  $C_p = \left[ (K + (4/3)G)/\rho \right]^{1/2}$ .

Using this equation, the applied maximum velocity is found to be approximately 0.04 m/sec. [Figure 1.5](#) shows a peak velocity of approximately 0.06 m/sec. Differences between estimated velocities and measured velocities result from using the intact rock modulus instead of the equivalent deformation modulus, which takes into account the joint deformation.




**Figure 1.5** Plot of y-velocity at top of model (applied stress =  $1.25 \times \cos(2\pi 100t)$ )


In the second example, a stress wave with peak stress of 12.5 MPa (“effective” stress = 6.25 MPa) was applied. The stress distribution in the roof of the excavation after 0.02 second is shown in [Figure 1.6](#). This figure shows that the roof block is unstressed, indicating that the block has loosened. Displacement versus time plots ([Figure 1.7](#)) also indicate that the block has loosened and is falling. As a matter of interest, the problem geometry and stress distribution at three later times are presented in [Figures 1.8](#) through [1.10](#).

The predicted velocity (from the equation above) at the top of the problem is 0.4 m/sec. The velocity calculated from the model is shown in [Figure 1.11](#). Again, differences between predicted and measured velocities result from using intact rock modulus instead of rock mass deformation modulus.


[Section 1.3](#) contains a listing of the data file for this model.




**Figure 1.6** Stress distribution in roof of excavation after 0.02 sec. (applied stress =  $12.5 \times \cos(2\pi 100t)$ )




**Figure 1.7** y-displacement histories for two points on excavation boundary (applied stress =  $12.5 \times \cos(2\pi 100t)$ )




**Figure 1.8** Stress distribution around excavation after 0.25 sec. (applied stress =  $12.5 \times \cos(2\pi 100t)$ )



**Figure 1.9** Stress distribution around excavation after 0.50 sec. (applied stress =  $12.5 \times \cos(2\pi 100t)$ )



**Figure 1.10** Stress distribution around excavation after 0.75 sec. (applied stress =  $12.5 \times \cos(2\pi 100t)$ )



**Figure 1.11** Plot of y-velocity at top of model (applied stress =  $12.5 \times \cos(2\pi 100t)$ )



### 1.3 Listing of Data File

#### *Example 1.1 SEISMIC.DAT*

---

```

;File:seismic.dat
model Title 'SEISMIC INDUCED ROOF COLLAPSE'
model new
block tolerance corner-round-length 0.01
; define original boundary of modeled region
block create polygon -25 -20 -25 20 25 20 25 -20
; generate joint pattern over entire original region
block cut joint-set angle 45 spacing 5 origin 0 0
block cut joint-set angle 351 spacing 5 origin 0 0
; put in joints needed for the later excavation
block cut crack -5.0 -2.5 5.0 -2.5
block cut crack -5.0 2.5 5.0 2.5
block cut crack -5 -2.5 -5 2.5
block cut crack 5 -2.5 5 2.5
block cut crack 2.25 2.5 1.93 5
; generate fdef zones and assign block and joint properties
;block zone gen edge 30
block zone gen edge 9.0 range pos-x -30 30 pos-y -30 30
block zone group 'mat1'
block zone cmodel assign elastic density 0.003 bulk 3.906E4 ...
    shear 3.178E4 range group 'mat1'
block contact group 'jmat1'
block contact cmodel assign area stiffness-shear 2E4 ...
    stiffness-normal 2E4 friction 30 range group 'jmat1'
; new contact default
block contact cmodel default area stiffness-shear 2E4 ...
    stiffness-normal 2E4 friction 30
; apply boundary conditions and initial conditions to
; consolidate model under field stresses
block edge apply stress -24.0 0.0 -24.0 ...
    gradient-x 0.0 0.0 0.0 gradient-y -0.03 0.0 -0.03
block insitu stress -24.0 0.0 -24.0 ...
    gradient-x 0.0 0.0 0.0 gradient-y -0.03 0.0 -0.03
block gridpoint apply vel-y 0 range pos-x -26 26 pos-y -21 -19
model gravity 0 -10
; track the x-displacement, and y-displacement over time
block gridpoint history displacement-x 0.0 7.0
block gridpoint history displacement-y 0.0 7.0
block solve ratio 1.0E-5
model save 'seismic1.sav'
; make excavation
block delete range pos-x -5 5 pos-y -2.5 2.5

```

---

```

block solve ratio 1.0E-5
model save 'seismic2.sav'
;
;
;   apply seismic load from top (peak velocity=0.04 m/sec)
;
; set up nonreflecting boundary
block gridpoint apply viscous-x range pos-x -26 -23 pos-y -21 21
block gridpoint apply viscous-x range pos-x 23 26 pos-y -21 21
block gridpoint apply viscous-y range pos-x -26 26 pos-y -21 -19
block gridpoint apply viscous-y range pos-x -26 26 pos-y 19 21
block edge apply property bulk 39060.0 shear 31780.0 density 0.0030
; apply sinusoidal stress wave
block edge apply stress 0.0 0.0 -1.25 history-y cosine 100 1.95E-2 ...
    range pos-x -26 26 pos-y 19 21
hist reset
block zone init rotation 0
block apply rotation 0
block mechanical time 0
block gridpoint init displacement-x 0
block gridpoint init displacement-y 0
history interval 1
block gridpoint history displacement-y -4.56 2.57
block gridpoint history displacement-y 0.0 2.57
block gridpoint history velocity-y 0.0 2.57
block gridpoint history velocity-y 4.0 2.57
block gridpoint history velocity-y -4.48 2.57
block gridpoint history velocity-y 0.0 20.0
block gridpoint history velocity-y 25.0 10.0
block gridpoint history velocity-y 25.0 -10.0
block gridpoint history velocity-y 0.0 -20.0
block gridpoint history velocity-y -25.0 -10.0
block gridpoint history velocity-y -25.0 10.0
block zone history stress-xx 25.0 10.0
block zone history stress-xx 25.0 -10.0
block zone history stress-xx -25.0 -10.0
block zone history stress-xx -25.0 10.0
block zone history stress-yy 0.0 20.0
block mechanical history time-total
;
block mechanical damping 0.1 1.0 mass
; 0.02 sec.
block cycle time 0.02
model save 'seismic3.sav'
;
model restore 'seismic2.sav'

```

```
; apply seismic load from top (peak velocity=0.4 m/sec)
; set up nonreflecting boundary
block gridpoint apply viscous-x range pos-x -26 -23 pos-y -21 21
block gridpoint apply viscous-x range pos-x 23 26 pos-y -21 21
block gridpoint apply viscous-y range pos-x -26 26 pos-y -21 -19
block gridpoint apply viscous-y range pos-x -26 26 pos-y 19 21
block edge apply property bulk 39060.0 shear 31780.0 density 0.0030
; apply sinusoidal stress wave
hist reset
block edge apply stress 0.0 0.0 -12.5 history-y cosine 100 1.95E-2 ...
    range pos-x -26 26 pos-y 19 21
block zone init rotation 0
block apply rotation 0
block mechanical time 0
block gridpoint init displacement-x 0
block gridpoint init displacement-y 0
block gridpoint history displacement-y -4.56 2.57
block gridpoint history displacement-y 0.0 2.57
block gridpoint history velocity-y 0.0 2.57
block gridpoint history velocity-y 4.0 2.57
block gridpoint history velocity-y -4.48 2.57
block gridpoint history velocity-y 0.0 20.0
block gridpoint history velocity-y 25.0 10.0
block gridpoint history velocity-y 25.0 -10.0
block gridpoint history velocity-y 0.0 -20.0
block gridpoint history velocity-y -25.0 -10.0
block gridpoint history velocity-y -25.0 10.0
block zone history stress-xx 25.0 10.0
block zone history stress-xx 25.0 -10.0
block zone history stress-xx -25.0 -10.0
block zone history stress-xx -25.0 10.0
block zone history stress-yy 0.0 20.0
block mechanical damping 0.1 1.0 mass
model save 'seismic4.sav'
;
model restore 'seismic4.sav'
; 0.02 sec.
hist interval 1
block cycle time 0.02
model save 'seismic5.sav'
;
; 0.25 sec.
block cycle time 0.23
model save 'seismic6.sav'
;
; 0.50 sec.
```

```
block cycle time 0.25
model save 'seismic7.sav'
;
; 0.75 sec
block cycle time 0.25
model save 'seismic8.sav'
```

---