

10 Influence of the Placement of Backfill in a Deep Longwall Excavation

10.1 Problem Statement

The use of the null, strain-softening and double-yield models is demonstrated in this simulation in which (1) the cohesion of the rock sharply decreases when failure occurs during staged excavation of a seam, and (2) the backfill yields in compression.

A five-meter thick seam is excavated in a bedded rock mass that is cut by horizontal joints with a vertical spacing of 5 m. Using symmetry, only half the seam thickness (2.5 m) is represented by the analysis. Four (4) horizontal joints are located at elevations of 2.5 m, 7.5 m, 12.5 m and 17.5 m above the center of the seam. [Figure 10.1](#) shows the *UDEC* model for this geometry.

The analysis includes five excavation steps. Fictitious vertical joints are placed every 10 m in the seam to model the phased excavation. After the first 10 m section (5 m due to symmetry) is excavated, it is then backfilled, and the second section is excavated. This pattern is repeated for the five steps.




Figure 10.1 *UDEC model of deep longwall excavation*

The seam and surrounding rock mass are assigned the same properties:

density	2700 kg/m ³
bulk modulus	38.9 GPa
shear modulus	29.9 GPa
friction angle	45°
dilation	0
cohesion	20 MPa at zero plastic shear strain, decreasing linearly to 0 at 3% plastic shear strain

The joint properties are

normal stiffness	100 GPa / m
shear stiffness	10 GPa / m
friction angle	30°
cohesion	0.05 MPa
tensile strength	0

The backfill properties are

density	1000 kg/m ³
bulk modulus	0.45 GPa
shear modulus	0.6 GPa
friction angle	40°
dilation	5°
cohesion	0

At zero plastic volumetric strain, the backfill material yields in compression at a mean stress of 10 kPa. The compressive strength then increases with increasing volumetric strain, following the curve given in [Figure 10.2](#).

The initial stress state before any excavation is

$$\begin{aligned}\sigma_v &= -67.5 \text{ MPa} \\ \sigma_h &= -33.8 \text{ MPa}\end{aligned}$$




Figure 10.2 *Cap pressure variation with plastic volumetric strain*

10.2 UDEC Analysis

The purpose of this analysis is to investigate the response of the rock mass and the behavior of the backfill during staged excavation. The seam and immediate roof use a finer zoning (Figure 10.3), in order to obtain a clearer picture of the stress distribution in the rock and backfill as the excavation and backfilling progresses.




Figure 10.3 Close-up view of the zoning for the 2.5-m thick seam and the roof

The *UDDEC* analysis begins at an equilibrium state defined by the initial stress conditions. The base boundary of the model is a “no vertical displacement” boundary (line of symmetry), while the two vertical boundaries have imposed zero horizontal displacement (also lines of symmetry). Note that, in this example, the boundaries are too close to permit the accurate simulation of a single longwall excavation. The example is only intended to demonstrate the behavior of the various block constitutive models.

The seam is excavated by changing the seam blocks to a null material (**block zone cmodel assign null**), and backfilled by changing the null material to a double-yield material (**block zone cmodel assign double-yield**). The excavation phase begins by excavating (nulling) a 10-m long section of blocks at the lower-left corner of the model. After a 10-m section has been excavated, backfill is placed in the section while the next section is excavated. Note that the excavation and backfilling are performed instantaneously in this example. Figure 10.4 shows the material models assigned for the second excavation/fill step.




Figure 10.4 *Second excavation (block zone cmodel assign null) step and first fill (block zone cmodel assign double-yield) step*

The vertical displacements in the roof, and the vertical stress in the backfill, are monitored while the excavation continues.

The stress states, after the seam has been excavated 10 m, 20 m, 30 m, 40 m and 50 m, are shown in [Figures 10.5](#) through [10.9](#). The distribution of the vertical stress in the backfill is superimposed on the stress plots. These figures give a broad picture of the system response. A classical arching effect develops.

Very far from the face, the vertical stress sustained by the backfill must approach the in-situ stress (67.5 MPa). It can be seen in [Figure 10.9](#) (40 m behind the face excavated for 50 m) that the backfill is sustaining only about one-third of the load. If a Mohr-Coulomb material had been used in place of the double-yield model, the backfill would have already taken up most of the load (compare [Figure 10.9](#) to [Figure 10.10](#)). The volumetric failure produced by the double-yield model results in the reduced load in the backfill.




Figure 10.5 Stress state, 10 m excavated




Figure 10.6 Stress state, 20 m excavated and 10 m backfilled




Figure 10.7 Stress state, 30 m excavated and 20 m backfilled




Figure 10.8 Stress state, 40 m excavated and 30 m backfilled




Figure 10.9 Stress state, 50 m excavated and 40 m backfilled




Figure 10.10 Stress state, 50 m excavated and 40 m backfilled with Mohr-Coulomb material

10.3 Listing of Data File

Example 10.1 LONGWALL.DAT

```

model new
;File:longwall.dat
model Title 'Backfill in Longwall Excavation'
block config
block tolerance corner-round-length 1E-2
block tolerance minimum-edge-length 2E-2
block contact tolerance overlap 1.0
block create polygon 0 0 0 50 80 50 80 0
block cut crack 0 2.5 80 2.5
block cut crack 0 7.5 80 7.5
block cut crack 0 12.5 80 12.5
block cut crack 10 0 10 2.5
block cut crack 20 0 20 2.5
block cut crack 30 0 30 2.5
block cut crack 40 0 40 2.5
block cut crack 50 0 50 2.5
block cut crack 60 0 60 2.5
block cut crack 70 0 70 2.5
block cut crack 0 3.5 80 3.5
block zone gen quad 1.95 0.5 range pos-x 0 80 pos-y 0 2.5
block zone gen quad 1.95 1.0 range pos-x 0 80 pos-y 2.5 3.5
block zone gen quad 4.0 2.0 range pos-x 0 80 pos-y 3.5 7.5
block zone gen quad 8.0 5.0 range pos-x 0 80 pos-y 7.5 17.5
block zone gen quad 16.0 16.5 range pos-x 0 80 pos-y 17.5 50
; prop mat 1 needed to prevent the null zones from collapsing
block property material 1 bulk 4e10 shear 3e10
block zone group 'rock'
block zone cmodel assign strain-softening density 2.7E3 bulk 3.89E10 ...
    shear 2.99E10 friction 45 cohesion 2E7 table-cohesion 1 tension 5e4 ...
    range group 'rock'
table '1' delete
table '1' add 0 2E7 3E-2 0
block contact group 'joint'
block contact cmodel assign area stiffness-shear 1E10 ...
    stiffness-normal 1E11 friction 30 cohesion 5E4 range group 'joint'
; new contact default
block contact cmodel default area stiffness-shear 1E10 ...
    stiffness-normal 1E11 friction 30 cohesion 5E4
block insitu stress -3.38E7 0.0 -6.75E7 stress-ZZ -5.0E7
block edge apply stress 0.0 0.0 -6.75E7 ...
    range pos-x -1E-2 80.01 pos-y 49.99 50.01
block gridpoint apply velocity-x 0 range pos-x -0.1 0.1 pos-y -0.1 50.01

```

```

block gridpoint apply velocity-x 0 range pos-x 79.99 80.01 pos-y -0.1 50.01
block gridpoint apply velocity-y 0 range pos-x -0.1 80.01 pos-y -0.1 0.1
block zone history stress-yy 10.05 2.2
block zone history stress-yy 20.05 2.2
block zone history stress-yy 30.05 2.2
block zone history stress-yy 40.05 2.2
block zone history stress-yy 50.05 2.2
block gridpoint history displacement-y 5.0 2.5
block gridpoint history displacement-y 15.0 2.5
block gridpoint history displacement-y 25.0 2.5
block gridpoint history displacement-y 35.0 2.5
block gridpoint history displacement-y 45.0 2.5
block solve ratio 1.0E-5 elastic
model save 'fill0.sav'

model restore 'fill0.sav'
block zone group 'Null:chamber 1' range atblock 5 2.25
block zone cmodel assign null range group 'Null:chamber 1'
;
block solve ratio 1.0E-5
model save 'fill1.sav'

table '2' delete
table '2' add 0,1E4 2E-2,2E5 4E-2,8E5 6E-2,1.3E6 0.1,5E6 ...
    0.12,8.5E6 0.14,1.15E7 0.16,1.9E7 0.18,3.4E7 0.2,5E7 1e10,5e7
;
block zone cmodel assign double-yield density 1E3 bulk-maximum 4.5E8 ...
    shear 6E8 friction 40 dilation 5 ...
    pressure-cap 1E4 table-pressure-cap 2 ...
    range group 'Null:chamber 1'
;
block zone group 'backfill' range group 'Null:chamber 1'
block zone group 'Null:chamber 2' range atblock 15.0 1.0
block zone cmodel assign null range group 'Null:chamber 2'
block solve ratio 1.0E-5
model save 'fill2.sav'
;
block zone cmodel assign double-yield density 1E3 bulk-maximum 4.5E8 ...
    shear 6E8 friction 40 dilation 5 ...
    pressure-cap 1E4 table-pressure-cap 2 range ...
    group 'null:chamber 2'
block zone group 'backfill' range group 'Null:chamber 2'
;
block zone cmodel assign double-yield density 1E3 bulk-maximum 4.5E8 ...
    shear 6E8 friction 40 dilation 5 ...
    pressure-cap 1e4 table-pressure-cap 2 ...

```

```
    range group 'backfill'
;
block zone group 'Null:chamber 3' range atblock 25.0 1.0
block zone cmodel assign null range group 'Null:chamber 3'
block solve ratio 1.0E-5
model save 'fill3.sav'
;
block zone cmodel assign double-yield density 1E3 bulk-maximum 4.5E8 ...
    shear 6E8 friction 40 dilation 5 ...
    pressure-cap 1E4 table-pressure-cap 2 ...
    range group 'null:chamber 3'
block zone group 'backfill' range group 'Null:chamber 3'
;
block zone cmodel assign double-yield density 1E3 bulk-maximum 4.5E8 ...
    shear 6E8 friction 40 dilation 5 ...
    pressure-cap 1e4 table-pressure-cap 2 ...
    range group 'backfill'
;
block zone group 'Null:chamber 4' range atblock 35.0 1.0
block zone cmodel assign null range group 'Null:chamber 4'
block solve ratio 1e-5
model save 'fill4.sav'
;
block zone cmodel assign double-yield density 1E3 bulk-maximum 4.5E8 ...
    shear 6E8 friction 40 dilation 5 ...
    pressure-cap 1E4 table-pressure-cap 2 ...
    range group 'null:chamber 4'
block zone group 'backfill' range group 'Null:chamber 4'
;
block zone cmodel assign double-yield density 1E3 bulk-maximum 4.5E8 ...
    shear 6E8 friction 40 dilation 5 &
    pressure-cap 1e4 table-pressure-cap 2 ...
    range group 'backfill'
;
block zone group 'Null:chamber 5' range atblock 45.0 1.0
block zone cmodel assign null range group 'Null:chamber 5'
block solve ratio 1e-5
model save 'fill5.sav'
ret
```
