

## 12 Blocks Bouncing down Slope

### 12.1 Problem Statement

Engineers are sometimes required to design slopes that must prevent loose rocks from falling onto roadways. The slopes are designed with benches or berms intended to catch falling debris. The geometry of these measures is difficult to determine by computer modeling. The path and final resting position of a falling block is highly dependent on size, shape, location and the properties of the surface on which it bounces. This is an example of a chaotic system where the final result is extremely sensitive to initial conditions. The final result is also sensitive to the timestep in *UDEC*. *UDEC* will update the timestep at each **block step** or **block cycle** command, so the results may vary if multiple **block cycle** commands are used during an analysis.\* It is often the case that this type of analysis is done by varying the input parameters and presenting a statistical result. The cell space contact-detection logic in *UDEC* can be used to look at problems where there may be many flying blocks.

The geometry in this case is shown in [Figure 12.1](#). The blocks in this example are for demonstration purposes, and are much larger than would normally be used in this type of analysis. The blocks in this model are modeled as rigid blocks. All properties in this analysis are assumed. While the density of the blocks is fairly easy to determine, the stiffnesses of the joints may be difficult to obtain. A range in values should be used to determine the range of results.

The properties of rock and joints are summarized:

#### *Rock*

density	2500 kg/m <sup>3</sup>
---------	------------------------

#### *Joints*

normal stiffness	5 GPa/m
shear stiffness	5 GPa/m
friction	30°

---

\* If only point contacts (**block contact cmodel assign point**) are used, the timestep will remain constant.

## 12.2 UDEC Analysis

The first task in this type of analysis is to determine the damping parameters that will give an appropriate surface behavior. The bounce behavior of a block is often referred to as coefficient of rebound, and is the ratio of the height of the bounce to the height of the drop.

$$\begin{aligned} Cr &= Hr/Hd \\ Cr &= \text{coefficient of rebound} \\ Hr &= \text{height of rebound} \\ Hd &= \text{height of drop} \end{aligned}$$

In *UDEC*, the rebound can be controlled by manipulation of the damping parameters. Unfortunately, the damping parameters are global constants, so the rebound will vary for different size blocks. The damping should be set for the average block size. In this example, most blocks are 1.0 m square. Assume for this case that  $Cr$  is 0.5. The natural frequency of the system is

$$\text{frequency} = \frac{1}{2\pi} \sqrt{\frac{kl}{m}} = 225 \text{ cycles/second}$$

where  $l$  = joint length (1.0 m, in this case)

$k$  = joint stiffness (5 GPa, in this case)

$m$  = mass of upper block (2500 kg)

Using this frequency and dropping a single block from a height of 1.0 m, the fraction of critical damping required to get a rebound of 50% is 0.23. Therefore, the command

```
damp 0.23 225 stiff
```

will result in a coefficient of rebound of .5.

[Figure 12.2](#) shows the position and velocities after 25,000 cycles, and [Figure 12.3](#) shows the blocks and velocities after 50,000 cycles. It is clear that, for the parameters chosen in this case, the bench is not sufficient to catch the blocks falling down the slope.

The data file for this example is listed in [Section 12.3](#). The file also demonstrates the *UDEC* commands to create a movie of the bouncing blocks.




Figure 12.1 UDEC model of blocks to be dropped onto the slope





Figure 12.2 Position and velocities of the blocks after 25,000 cycles



**Figure 12.3** Position and velocities of the blocks after 50,000 cycles

### 12.3 Listing of Data File

#### *Example 12.1 BOUNCE.DAT*

---

```

; File:bounce.dat
modell title 'Blocks Bouncing Down Slope'
; cell logic example : slope rock fall with rigid blocks
;
model new
block config cellspace
block tolerance corner-round-length 0.01
block create polygon 0 0 0 15 3 10 8 9 11 2 15 2 15 0
block create polygon 2 12 2 15 4 15 4 12
block cut crack 2 13 4 13
block cut crack 2 13 3 15
block cut crack 3 12 3 15
block cut crack 2 14 4 14
;
block change material 1
block property material 1 density 2.55e-3
block contact group 'area contact'
block contact cmodel assign area st-n 5E3 st-s 5E3 friction 30 ...

```

---

```
    range group 'area contact'
; new contact default
block contact cmodel default area st-n 5E3 st-s 5E3 friction 30
;
block fix range pos-x 0 15 pos-y 0 10
block mechanical gravity 0 -10
;
block mechanical damp .23 225.0 stiff
block contact tolerance overlap .1
;
model save 'b1.sav'
block cycle 100
fish define check_extent
    command
        block delete range pos-x 15 30
    end_command
end
;
fish define do_steps
    loop i (1,5)
        check_extent
        command
            block cycle 5000
        end_command
    end_loop
end
@do_steps
model save 'c25k.sav'
@do_steps
model save 'c50k.sav'
@do_steps
@do_steps
model save 'bounce.sav'
return
```

---

