

9 Flow from a Borehole in a Biaxial Stress Field

9.1 Problem Statement

A borehole is located in a rock mass containing two orthogonal sets of joints and subjected to a biaxial in-situ stress state. Fluid is injected into the borehole at a constant flow rate. The purpose of the analysis is to evaluate the influence of the in-situ stress state on flow into the joints.

The in-situ stress field is

$$\sigma_{xx} = -25 \text{ MPa}$$

$$\sigma_{yy} = -20 \text{ MPa}$$

$$p = 10 \text{ MPa}$$

in which σ_{xx} and σ_{yy} are *total* stresses in the x - and y -directions, and p is the fluid pressure in the joints.

The intact rock block properties are

$$\text{density} \quad 2500 \text{ kg/m}^3$$

$$\text{bulk modulus} \quad 66.667 \text{ GPa}$$

$$\text{shear modulus} \quad 40.0 \text{ GPa}$$

The joint spacing is 2.5 m. The joint properties are

$$\text{normal stiffness} \quad 2 \times 10^5 \text{ MPa/m}$$

$$\text{shear stiffness} \quad 2 \times 10^5 \text{ MPa/m}$$

$$\text{permeability factor} \quad 300 \text{ m} \cdot \text{sec/kg}$$

$$\text{residual hydraulic aperture} \quad 2 \times 10^{-5} \text{ m}$$

$$\text{aperture at zero normal stress} \quad 1 \times 10^{-4} \text{ m}$$

9.2 UDEC Analysis

The *UDEC* model for this problem is shown in [Figure 9.1](#). The data file is listed in [Section 9.3](#).

For this transient flow analysis, the flow is assumed to be incompressible (**block fluid incompressible**). Fluid is injected at a constant flow rate of $10^{-3} \text{ m}^3/\text{sec}$ into the borehole via the **block domain well** command.

Given the difference in in-situ stresses, the dominant flow is in the horizontal joint set. [Figure 9.2](#) shows the flow pattern after 15 seconds. The change in hydraulic aperture after 15 seconds is illustrated by the aperture plot in [Figure 9.3](#).

The fluid pressure histories at five points in the model (see [Figure 9.1](#)) are plotted in [Figure 9.4](#). Points 1 and 3, located on the horizontal joint, indicate a faster response than do points 2 and 4, located on the vertical joint.




Figure 9.1 *UDEC model of a borehole in a rock mass containing two orthogonal joint sets*




Figure 9.2 Flow in joints at 15 seconds after fluid flow initiates from the borehole




Figure 9.3 Joint hydraulic apertures at 15 seconds after fluid flow initiates from the borehole




Figure 9.4 Fluid pressure histories at points 1, 2, 3, 4 and 5

9.3 Listing of Data File

Example 9.1 BH.DAT

```

model new
;File:bh.dat
; -----
;
; pressurized borehole --- biaxial stress field
;
;         insitu sxx=-25 syy=-20 pp=10
;
;         borehole radius = 0.2
;
; -----pri dom-----
model title "Flow from a Borehole in a Biaxial Stress Field"
block config fluid
block fluid clear incompressible off
block tolerance corner-round-length 1E-3
block tolerance minimum-edge-length 2E-3
block create polygon -10 -10 -10 10 10 10 10 -10
block cut joint-set angle 0 spacing 2.5 origin 0 0
block cut joint-set angle 90 spacing 2.5 origin 0 0
block cut arc 0 0 0.2 0 360 8 join
block delete range annulus center 0 0 radius 0 0.2
;
block zone gen edge 10.0
;
block zone group 'rock'
block zone cmodel assign elastic density 2.5E-3 bulk 6.667E4 shear 4E4 ...
    range group 'rock'
block contact group 'joint'
block contact cmodel assign area stiffness-shear 2E5 ...
    stiffness-normal 2E5 friction 30 permeability-factor 3E8 ...
    aperture-residual 2E-5 aperture-zero-load 1E-4 range group 'joint'
; new contact default
block contact cmodel default area stiffness-shear 2E5 ...
    stiffness-normal 2E5 friction 30 permeability-factor 3E8 ...
    aperture-residual 2E-5 aperture-zero-load 1E-4
;
block insitu stress -25.0 0.0 -20.0 pore-pressure 10
block gridpoint apply velocity-x 0 range pos-x -10.1 -9.9 pos-y -10.1 10.1
block gridpoint apply velocity-x 0 range pos-x 9.9 10.1 pos-y -10.1 10.1
block gridpoint apply velocity-y 0 range pos-x -10.1 10.1 pos-y -10.1 -9.9
block gridpoint apply velocity-y 0 range pos-x -10.1 10.1 pos-y 9.9 10.1
block solve ratio 1.0E-5 elastic

```

```
model save 'bh0.sav'
;
hist reset
block mechanical time 0
history interval 1
block domain history pore-pressure -5.0 0.0
block domain history pore-pressure 0.0 -5.0
block domain history pore-pressure -9.0 0.0
block domain history pore-pressure 0.0 -9.0
block domain history pore-pressure 0.0 0.0
block fluid history unbalanced-volume
block mechanical history time-total
;
block fluid flow on
block edge apply pore-pressure 10.0
;pfree range pos-x -0.5,0.5 pos-y -0.5,0.5
block domain well flow 0.0010 atdomain 0 0
block fluid incompressible substep-mechanical 1000
block fluid incompressible tolerance-volume 1.0E-4
block fluid incompressible timestep 0.5
block cycle ftime 15.0
model save 'bh1.sav'
```
