


### 3 Tunnel Support Loading

#### 3.1 Problem Statement

This simulation demonstrates the application of *UDEC* to examine lined tunnels, with specific emphasis on loads developed in the concrete liners. This example also illustrates the procedure to model the individual stages of a sequential construction operation.

The idealized geometry of the tunnel system is shown in [Figure 3.1](#). The system consists of two tunnels on 12-meter centers at a depth of roughly 70 meters (centerline) beneath the sea bed. The water level is initially 110 meters above the tunnel centerline. The small (service) tunnel is 5.24 meters in diameter, with a 37 cm-thick concrete liner. The main tunnel is 8.22 meters in diameter, with a 46 cm-thick concrete liner. The service tunnel is driven and lined prior to excavation and lining of the main tunnel. After installation of the main tunnel lining, the water level is raised an additional 100 meters.



**Figure 3.1** *Idealized geometry of service tunnel and main tunnel*

The sequence of construction activities is:

- (1) excavation of the service tunnel;
- (2) lining of the service tunnel;
- (3) excavation of the main tunnel;
- (4) lining of the main tunnel; and
- (5) raising of the water level.

The objective of the analysis is to evaluate the tunnel support for the service tunnel and the main tunnel at each construction stage.

The material properties for this example are listed below.

*Rock* – The tunnels are excavated in rock that has the following material properties:

elastic modulus	0.89 GPa
Poisson's ratio	0.35
uniaxial compressive strength	3.5 MPa
cohesion	1 MPa
density	1340 kg/m <sup>3</sup>

*Concrete Liner* – The elastic modulus for the concrete liner is 24 GPa, and the Poisson's ratio is 0.19. The liner is assumed to behave as a linear-elastic material.

### 3.2 UDEC Analysis

The *UDEC* model created for this analysis is shown in [Figure 3.2](#). The dimensions are selected so that the centerline of the tunnels corresponds to the location  $y = -70$  m. Note that the model boundaries are very close to the tunnel excavations. This model is only intended to provide fast calculations, for demonstration purposes. A larger model would be required for practical solutions.

Construction joints are used to create the tunnel boundaries by adding the **join** keyword to the **block cut crack** and **block cut arc** commands. In this way, it is not necessary to assign joint models and properties for this example.

The lower and side boundaries of the *UDEC* model are fixed with rollers. The weight of the sea water above the sea bed is simulated by applying the equivalent pressure of 30 m of water head to the top surface of the model. The tunnels are assumed to be lined with a waterproof liner. Thus there is no need to perform a transient groundwater flow analysis. The pore water pressure is accounted for by setting the unit weight of the rock to the submerged unit weight. The vertical-to-horizontal stress ratio is assumed to be hydrostatic.

For this example, the five construction activities listed above are simulated by three modeling stages. Excavation of the main tunnel and lining of the service tunnel are modeled as one activity that occurs instantaneously. Lining of the main tunnel and raising the water level are also assumed to occur instantaneously. For a more realistic simulation, these activities should be simulated separately by reducing the tractions around the tunnels gradually and installing the support after some prescribed relaxation takes place.

In the first modeling stage, after gravity stresses have been initialized in the body, the service tunnel is mined, and *UDEC* is cycled until equilibrium is achieved. The resulting elastic displacements are given in [Figure 3.3](#).




Figure 3.2 UDEC model zoning with service tunnel excavated






Figure 3.3 Elastic displacements due to excavation of service tunnel

In the second stage, the service tunnel is lined and the main tunnel is excavated. Sixteen beam elements are used to model the concrete liner for the service tunnel. [Figures 3.4](#) and [3.5](#) illustrate the displacements and principal stress distribution resulting from excavation of the main tunnel. Note that the entire service tunnel translates toward the main tunnel.


In the third stage, an additional load is applied to the top of the model to simulate the weight of an additional 100 meters of water. The **block struct liner apply pressure** command is used to apply hydrostatic loads (to the tunnel liner) representing the water table 210 m above the tunnel center line. [Figures 3.6](#) and [3.7](#) display the thrust and moment distributions, respectively, that develop after the additional load is applied. The maximum thrust in the service tunnel is approximately  $5.5 \times 10^6$  N, and that in the main tunnel is approximately  $9.0 \times 10^6$  N.




**Figure 3.4** *Displacements after mining of main tunnel*



**Figure 3.5** *Principal stress distribution after mining of main tunnel*



**Figure 3.6** Thrust in liners after water level is raised



**Figure 3.7** Moment in liners after water level is raised

### 3.3 Listing of Data File

#### *Example 3.1 TUNNEL.DAT*

---

```

model new
;File:tunnel.dat
Model title 'Tunnel support'
block tolerance corner-round-length 0.1
block create polygon 0 -90 0 -30 69 -30 69 -90
block cut crack 0 -70 60 -70 join
block cut crack 30 0 30 -90 join
block cut crack 42 0 42 -90 join
block cut arc 30 -70 34.11 -70 360 12 join
block cut arc 42 -70 44.62 -70 360 8 join
block cut arc 30 -70 35.5 -70 360 12 join
block cut arc 42 -70 47.5 -70 360 12 join
block zone gen edge 2.0
; rock properties
block zone group 'rock'
block zone cmodel assign mohr-c density 1.34E3 bulk 9.9E8 shear 3.3E8 ...
    friction 30 cohes 1E6 tens 1e5 range group 'rock'
; initial stress state
block edge apply stress 102000.0 0.0 102000.0 ...
    gradient-y 13400.0 0.0 13400.0
block insitu stress 102000.0 0.0 102000.0 ...
    gradient-y 13400.0 0.0 13400.0 stress-ZZ 102000.0 gradient-z 0.0 13400.0
;boundary stress -102000.0,0.0,-102000.0
;insitu stress -102000.0,0.0,-102000.0 &
;      szz -102000.0
block gridpoint apply velocity-x 0 range position-x -1 1
block gridpoint apply velocity-x 0 range position-x 68 70
block gridpoint apply velocity-y 0 range position-y -91 -89
model gravity 0.0 -10.0
block solve ratio 1.0E-5 elastic
model save 'tun1.sav'
;
; excavate service tunnel
block delete range annulus center 42 -70 radius 0 2.62
block gridpoint history displacement-y 42.0 -67.0
block gridpoint history displacement-y 42.0 -73.0
block gridpoint history displacement-x 39.0 -70.0
block gridpoint history displacement-x 45.0 -70.0
block contact reset displacement
block gridpoint init displacement-x 0
block gridpoint init displacement-y 0
block solve ratio 1.0E-5

```

---

```

model save 'tun2.sav'
;
; line service tunnel and excavate main tunnel
bl struct liner create by-end-points ...
  begin 39.5747 -69.5376 end 39.5747 -69.5376 ...
  length-maximum 100.0 length-minimum 0.2 material-beam 1
bl struct beam prop mat 1 density 2.4E3 poisson 0.2 ...
  yield-compression 1E10 yield-tension 1E10 young 2.4E10 ...
  cross-sectional-area 0.37 moi 4.221E-3 shape-factor 0.8333 ...
  spacing 1 thickness 0.37 width 1 coupling-cohesion 1E10 ...
  coupling-stiffness-normal 1E8 coupling-stiffness-shear 1E7
block delete range annulus center 30 -70 rad 0 4.11
block gridpoint init displacement-x 0
block gridpoint init displacement-y 0
block solve ratio 1.0E-5
model save 'tun3.sav'

; line main tunnel and raise water level
bl struct liner create by-end-points ...
  begin 26.3077 -68.4617 end 26.3077 -68.4617 ...
  length-maximum 1.0E9 length-minimum 0.2 material-beam 2
bl struct beam prop mat 2 density 2.43E3 poisson 0.2 ...
  yield-compression 1E10 yield-tension 1E10 young 2.4E10 ...
  cross-sectional-area 0.46 moi 8.111E-3 shape-factor 0.8333 ...
  spacing 1 thickness 0.46 coupling-cohesion 1E10 ...
  coupling-stiffness-normal 1E8 coupling-stiffness-shear 1E7
bl edge apply stress 0.0 0.0 -1000000.0 ...
  range pos-x -1 91 pos-y -31 -29
block structure liner apply pressure inside 0.0 outside 2060000.0
block gridpoint init displacement-x 0
block gridpoint init displacement-y 0
block solve ratio 1.0E-5
model save 'tun4.sav'

```

---